Supporting Information

Achieving pure spin effects by artifact suppression in methyl adiabatic relaxation experiments

Fa-An Chao, Domarin Khago, R. Andrew Byrd*

Structural Biophysics Laboratory, Center for Cancer Research, National Cancer Institute, Frederick, Maryland 21702-1201, United States.

Contact information: R. Andrew Byrd
(V): 301-846-1407
Email: byrdra@mail.nih.gov

Table S1. Methyl data for chemical exchange from all experiments

	Ube2g2		Ubde2g2:G2BR	
Methyl group	k_{ex} (s^{-1}), geoHARD	R_{ex} (s^{-1}), CPMG	k_{ex} (s^{-1}), geoHARD	R_{ex} (s^{-1}), CPMG
L9-CD1		3.5±0.2		3.1±2.8
L9-CD2	6000±31	13.7±0.4		
L16-CD1	9226±990	20.8±1.1		
L16-CD2		7.6±0.1		
L18-CD1	5092±1580			
I24-CD1	3774±88	3.7±0.2		
V25-CG2			340±101	2.1±2.1
L40-CD1				3.0±2.5
I41-CD1	5271±252		2148±374	
V53-CG2	1110±503			
I57-CD1	9989±1828			
L58-CD2			14673±3137	
L66-CD1			3474±1551	
I82-CD1	9936±3869			
L93-CD1	3644±741			
V113-CG2				3.8±3.2
V116-CG1	5153±1387			4.2±4.0
V116-CG2	2231±1044			
I119-CD1			4575±127	
L121-CD1		3.6±0.1		
L121-CD2			3860±1027	
V124-CG2	4759±633			
L127-CD1	33955±8167			
V159-CG1			4677±1174	
L165-CD1				2.4±2.1



Supporting figure 1. Experimental observation of the artifacts in methyl HMQC spectra due to the pulse imperfections of different π pulses. The Ile region of the HMQC spectra of the deuterated ASAP1 PH domain[1] (~14 kDa) is collected at 25 ˚C and 600 MHz, and the proton carrier center is set at 4.7 ppm using (a) a regular π pulse, (b) a composite pulse, 90_x-180_y-90_x, or (c) a composite pulse, 90_x-270_y-90_x with a RF field of 20 kHz.



Supporting figure 2. Simulated off-resonance effects on magnetization inversion of different π pulses, (a) a simple rectangular π pulse, (b) a composite pulse, 90_x-180_y-90_x, or (c) a composite pulse, 90_x-270_y-90_x, assuming a RF field of 10 kHz (pw90 = 25 us).



Supporting figure 3. The removal of transverse cross-correlation between the dipole-dipole interaction and chemical shift anisotropy using composite pulse decoupling in methyl-geoHARD experiments. The experiments are carried out using the deuterated ASAP1 ZA domain[2] (~32 kDa) at 5 ˚C and 850 MHz using the new pulse sequence (Fig. 1). The normalized transverse relaxation decays of a selected methyl group are plotted with and without the composite pulse decoupling. Previously[2], the initial slope was used to minimize the effects of multi-exponential decay. With efficient composite pulse decoupling, the relaxation becomes purely mono-exponential. This mono-exponential behavior was observed for all methyl groups in ASAP1 ZA and for the example of Ube2g2.



Supporting figure 4. Unperturbed longitudinal relaxation rates in methyl-geoHARD experiments using the composite pulse decoupling. The experiments are carried out using the deuterated ASAP1 ZA domain[2] (~32 kDa) at 5 ˚C and 850 MHz using the new pulse sequence (Fig. 1). The longitudinal relaxation rates are measured with (Figure S5,e) and without (Figure S5,b) the composite pulse decoupling. The correlation is excellent, indicating that the composite pulse does not perturb the proton-proton spin-flip mechanism. If the decoupling creates non-uniform/incomplete spin inversion, then the spin-flip mechanism is perturbed and the correlation degrades[2].



Supporting figure 5. A comparison of the different proton decoupling strategies examined for methyl geoHARD experiments. (a, b) The original experiment from figure 1 of (2). (c) Repeated π-pulse decoupling, Figure S6 of (2). (d) Single, hard π-pulse decoupling. (e) Single, composite π-pulse decoupling as in Figure 1 of this paper. The implementation of the symmetric π-pulse decoupling within the repetition N for panels (c) and (d) requires complex pulse programming. An example of the explicit timing is shown in Supporting figure 6 for $R_{1\rho}$.



Supporting figure 6. Implementation of the composite pulse decoupling for $R_{1\rho}$ in Supporting figure 5e. Only the region of the sequence for the dashed component of Figure 1b is shown. The pulse sequence that performs this complex timing is provided in this supporting information (below). Similar implementation occurs for the $R_{2\rho}$ experiment.



Supporting figure 7. Methyl geoHARD data for Ube2g2. (a) Location of the 13 methyl reporters in Ube2g2. (b) The two clusters of methyl groups that connect between the back side of Ube2g2, where G2BR binds, and the RING binding site. The cluster located at the junction between helix α1 and helix α2 (magenta) have a range of k_{ex} from 2000 s^{-1} to 9000 s^{-1}. The cluster located at the junction between helix α2 and β (dark magenta) have an average k_{ex} of ~4000 s^{-1}. (c) The bar representation of the k_{ex} rates of all 13 reported methyl groups in the free Ube2g2.

a



b



Supporting figure 8. CPMG data for Ube2g2. (a) Location of methyl groups that exhibited detectable relaxation dispersion in single-quantum CPMG experiments[3]. (b) The bar representation of the R_{ex} rates of all methyl groups with measurable R_{ex} rates in the free Ube2g2 using single-quantum CPMG experiments at 800 MHz. (The R_{ex} rates are approximated with the differences between $R_{2,eff}$ rates at 50 Hz and 1000 Hz of νCPMG frequencies. If the differences are smaller than 2 s^{-1} or smaller than the experimental errors, they will not be reported.)

a



b



Supporting figure 9. Methyl geoHARD for Ube2g2:G2BR. (a) Location of the 7 methyl reporters in Ube2g2:G2BR. (b) The bar representation of the k_{ex} rates of all 7 reported methyl groups in the Ube2g2 saturated with G2BR.

a



b



Supporting figure 10. CPMG data for Ube2g2:G2BR. (a) Location of methyl groups that exhibited detectable relaxation dispersion in single-quantum CPMG experiments[3]. (b) The bar representation of the R_{ex} rates of all methyl groups with measurable R_{ex} rates in the Ube2g2 saturated with G2BR using single-quantum CPMG experiments at 800 MHz. (The R_{ex} rates are approximated with the differences between $R_{2,eff}$ rates at 50 Hz and 1000 Hz of νCPMG frequencies. If the differences are smaller than 2 s^{-1} or smaller than the experimental errors, they will not be reported.)

References

1. Jian, X.; Tang, W. K.; Zhai, P.; Roy, N. S.; Luo, R.; Gruschus, J. M.; Yohe, M. E.; Chen, P. W.; Li, Y.; Byrd, R. A.; Xia, D.; Randazzo, P. A., Molecular Basis for Cooperative Binding of Anionic Phospholipids to the PH Domain of the Arf GAP ASAP1. *Structure* **2015,** *23* (11), 1977-88.
2. Chao, F. A.; Li, Y.; Zhang, Y.; Byrd, R. A., Probing the Broad Time Scale and Heterogeneous Conformational Dynamics in the Catalytic Core of the Arf-GAP ASAP1 via Methyl Adiabatic Relaxation Dispersion. *J Am Chem Soc* **2019,** *141* (30), 11881-11891.
3. Lundstrom, P.; Vallurupalli, P.; Religa, T. L.; Dahlquist, F. W.; Kay, L. E., A single-quantum methyl C-13-relaxation dispersion experiment with improved sensitivity. *J Biomol Nmr* **2007,** *38* (1), 79-88.

Pulse sequences developed in Bruker TopSpin 3.6.2

Shape files and analysis software are available from the authors and at NMRbox.org

---

$R_{1\rho}$ Adiabatic Relaxation Dispersion with composite pulse decoupling

```
;SQ methyl-HMQC written by Frank Chao for Adiabatic R1rho relaxation dispersion
;include composite pulse decoupling
;J Am Chem Soc. 2019 Jul 31;141(30):11881-11891.
;Chao, Khago, Byrd, J.BioNMR 2020 in press.


;HMQC detection
;pseudo 3D experiment
;multiplet editing for selection of SQ coherence and relaxation dispersion
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define list<loopcounter> cpmglist=<$VCLIST>    ;the N cycles of composite
adiabatic pulses

"p2=p1*2"
"p4=p3*2"
"d4=1.8m"
"d5=0.66666666m"
"d11=30m"
"d17=d16+p1*5"

"d0=3u"

"in0=inf2/2"

"DELTA1=d4-p19-d16-0.5*p4"
"DELTA2=d5-p20-d16-0.5*p4"
"DELTA4=d4-p19-d16-0.5*p4-8u-p11"
"p24=4000u"

"acqt0=0"
baseopt_echo
aqseq 312


1  ze
   d11 st0
   3m
2  3m
3  6m
4  6m
```

```
   "l4=cpmglist"
   "p62=(l5*p24*4*0.838)-(cnst61*l4*p24*4)"

if "l4 > 0"
   {
   if "l4 % 2 == 0"
     {
     "COUNTER=(l4-2)/2"
     }
   else
     {
     "COUNTER=(l4-1)/2"
     }
   }

   4u pl2:f2 pl1:f1

   d1
   4u UNBLKGRAD
   (p3 ph1):f2
   p16:gp1
   d16

   (p1 ph1)
   p19:gp2
   d16
   DELTA1
   (center (p2 ph1) (p4 ph1):f2 )
   DELTA1
   p19:gp2
   d16
   (p1 ph9)
   4u pl0:f1
   (p11:sp1 ph3:r):f1
   4u pl1:f1

   p16:gp3
   d16
   (p3 ph11):f2

   p20:gp4
   d16
   DELTA2
   (center (p2 ph3) (p4 ph3):f2 )
   DELTA2
   p20:gp4
   d16
   (p3 ph1):f2
   p19:gp5

if "l4 == 0"
   {
   d16
   (p1 ph2 p1*3 ph3 p1 ph2)
   }
else
   {
   d17
   }

   1u pl63:f2
```

```
if "l4 > 0"
   {
   if "l4 % 2 == 1"
     {
   5 2u
     (p24:sp24 ph3 4u p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9 4u p24:sp24 ph9):f2
     2u
     lo to 5 times COUNTER

     2u
     (p24:sp24 ph3):f2
     4u
     (center (p1 ph2 p1*3 ph3 p1 ph2):f1 (p24:sp24 ph3 4u p24:sp24 ph9):f2)
     4u
     (p24:sp24 ph9):f2
     2u

   6 2u
     (p24:sp24 ph3 4u p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9 4u p24:sp24 ph9):f2
     2u
     lo to 6 times COUNTER
     }
   else
     {
   7 2u
     (p24:sp24 ph3 4u p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9 4u p24:sp24 ph9):f2
     2u
     lo to 7 times COUNTER

     2u
     (p24:sp24 ph3 4u p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9):f2
     4u
     (center (p1 ph2 p1*3 ph3 p1 ph2):f1 (p24:sp24 ph9 4u p24:sp24 ph3):f2)
     4u
     (p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9 4u p24:sp24 ph9):f2
     2u

   8 2u
     (p24:sp24 ph3 4u p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9 4u p24:sp24 ph9):f2
     2u
     lo to 8 times COUNTER
     }
   }
   1u pl2:f2

   (p3 ph4):f2
   (p1 ph1)

   d0
```

```
   (p1 ph2 p1*3 ph3 p1 ph2)
   d0

   (p3 ph5):f2
   p19:gp7
   d16
   DELTA4
   4u pl0:f1
   (p11:sp1 ph8:r):f1
   4u pl1:f1
   (center (p2 ph1) (p4 ph1):f2 )
   4u pl0:f1
   (p11:sp1 ph8:r):f1
   4u pl1:f1
   DELTA4
   p19:gp7
   d16 pl12:f2 BLKGRAD
   goscnp ph31 cpd2:f2
   3m do:f2

   3m pl62:f2
   (p62 ph1):f2
   4u
   (p62 ph8):f2
   4u

   4u UNBLKGRAD
   p16:gp1*3
   d16
   (p1 ph1):f1
   4u pl0:f1
   (p11:sp1 ph8:r):f1
   4u pl1:f1
   p16:gp1*3
   d16
   4u BLKGRAD

   3m st cpmglist.inc
   lo to 2 times nbl

   3m cpmglist.res
   3m ipp4 ipp11 ipp31
   lo to 3 times ns

   6m mc #0 to 4
     F1QF()
     F2PH(calph(ph5, +90), caldel(d0, +in0))
exit


ph1=0
ph2=0
ph3=1
ph4=0 0 2 2
ph5=0

ph8=2
ph9=3
ph10=0
ph11=0 2
```

```
ph31=0 2 2 0


;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;pl62: f2 channel - a half of sp24 power level (in W)
;pl63: 0W
;sp1: f1 channel - shaped pulse 90 degree for water flip-back
;sp24: f2 channel - power level for 180 degree shaped pulse (adiabatic; 7kHz)
;spnam24: HS1_FC, HS2_FC, HS4_FC, or HS8_FC
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p11: f1 channel -  90 degree shaped pulse for water flip-back  [1 msec]
;p16: homospoil/gradient pulse (1ms)
;p19: homospoil/gradient pulse (0.5ms)
;p20: homospoil/gradient pulse (0.2ms)
;p24: f2 channel -  180 degree adiabatic pulse   [4 msec]
;d0 : incremented delay (2D)                  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                             [30 msec]
;d16: delay for homospoil/gradient recovery
;cnst61: scaling factor (HS1: 0.295, HS2: 0.513, HS4: 0.705, HS8: 0.836)
;l4: loop for R1rho delay. l4 may be 0,1,2,3,4,5
;l5: maximum value of l4
;inf2: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 4 * n
;ds: 16
;td1: number of experiments
;FnMODE: States-TPPI
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence



;for z-only gradients:
;gpz1: 10%
;gpz2: 6%
;gpz3: 50%
;gpz4: 10%
;gpz5: -20%
;gpz7: 25%


;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.50
;gpnam3: SMSQ10.100
;gpnam4: SINE.100
;gpnam5: SMSQ10.50
;gpnam7: SMSQ10.50


;$Id: SQ_methyl-HMQC_R1rho-3D.dc , 2020/04/02 F. Chao Exp $
```

$R_{2\rho}$ Adiabatic Relaxation Dispersion with composite pulse decoupling

```
;SQ methyl-HMQC written by Frank Chao for Adiabatic R2rho relaxation dispersion
;include composite pulse decoupling
;J Am Chem Soc. 2019 Jul 31;141(30):11881-11891.
;Chao, Khago, Byrd, J.BioNMR 2020 in press.


;HMQC detection
;pseudo 3D experiment
;multiplet editing for selection of SQ coherence and relaxation dispersion
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

define list<loopcounter> cpmglist=<$VCLIST>      ;the N cycles of composite
adiabatic pulses

"p2=p1*2"
"p4=p3*2"
"d4=1.8m"
"d5=0.66666666m"
"d11=30m"

"d0=3u"

"in0=inf2/2"

"DELTA1=d4-p19-d16-0.5*p4"
"DELTA2=d5-p20-d16-0.5*p4"
"DELTA4=d4-p19-d16-0.5*p4-8u-p11"
"p24=4000u"

"acqt0=0"
baseopt_echo
aqseq 312


1  ze
   d11 st0
   3m
2  3m
3  6m
4  6m

   "l4=cpmglist"
```

```
   "p62=(l5*p24*4*0.838)-(cnst61*l4*p24*4)"

if "l4 > 0"
   {
   if "l4 % 2 == 0"
     {
     "COUNTER=(l4-2)/2"
     }
   else
     {
     "COUNTER=(l4-1)/2"
     }
   }

   4u pl2:f2 pl1:f1

   d1
   4u UNBLKGRAD
   (p3 ph1):f2
   p16:gp1
   d16

   (p1 ph1)
   p19:gp2
   d16
   DELTA1
   (center (p2 ph1) (p4 ph1):f2 )
   DELTA1
   p19:gp2
   d16
   (p1 ph9)
   4u pl0:f1
   (p11:sp1 ph3:r):f1
   4u pl1:f1

   p16:gp3
   d16
   (p3 ph11):f2

   p20:gp4
   d16
   DELTA2
   (center (p2 ph3) (p4 ph3):f2 )
   DELTA2
   p20:gp4
   d16
   (p3 ph1):f2
   p19:gp5
   d16
   (p3 ph4):f2

   1u pl63:f2
if "l4 > 0"
   {
   if "l4 % 2 == 1"
     {
   5 2u
     (p24:sp24 ph3 4u p24:sp24 ph3):f2
     4u
     (p24:sp24 ph9 4u p24:sp24 ph9):f2
     2u
```

```
      lo to 5 times COUNTER

      2u
      (p24:sp24 ph3):f2
      4u
      (center (p1 ph2 p1*3 ph3 p1 ph2):f1 (p24:sp24 ph3 4u p24:sp24 ph9):f2)
      4u
      (p24:sp24 ph9):f2
      2u

   6 2u
      (p24:sp24 ph3 4u p24:sp24 ph3):f2
      4u
      (p24:sp24 ph9 4u p24:sp24 ph9):f2
      2u
      lo to 6 times COUNTER
      }
   else
      {
   7 2u
      (p24:sp24 ph3 4u p24:sp24 ph3):f2
      4u
      (p24:sp24 ph9 4u p24:sp24 ph9):f2
      2u
      lo to 7 times COUNTER

      2u
      (p24:sp24 ph3 4u p24:sp24 ph3):f2
      4u
      (p24:sp24 ph9):f2
      4u
      (center (p1 ph2 p1*3 ph3 p1 ph2):f1 (p24:sp24 ph9 4u p24:sp24 ph3):f2)
      4u
      (p24:sp24 ph3):f2
      4u
      (p24:sp24 ph9 4u p24:sp24 ph9):f2
      2u

   8 2u
      (p24:sp24 ph3 4u p24:sp24 ph3):f2
      4u
      (p24:sp24 ph9 4u p24:sp24 ph9):f2
      2u
      lo to 8 times COUNTER
      }
   }
   1u pl2:f2

if "l4 > 0"
   {
   (p1 ph1)
   }
else
   {
   (p1 ph8)
   }

   d0
   (p1 ph2 p1*3 ph3 p1 ph2)
   d0
```

```
   (p3 ph5):f2
   p19:gp7
   d16
   DELTA4
   4u pl0:f1
   (p11:sp1 ph8:r):f1
   4u pl1:f1
   (center (p2 ph1) (p4 ph1):f2 )
   4u pl0:f1
   (p11:sp1 ph8:r):f1
   4u pl1:f1
   DELTA4
   p19:gp7
   d16 pl12:f2 BLKGRAD
   goscnp ph31 cpd2:f2
   3m do:f2

   3m pl62:f2
   (p62 ph1):f2
   4u
   (p62 ph8):f2
   4u

   4u UNBLKGRAD
   p16:gp1*3
   d16
   (p1 ph1):f1
   4u pl0:f1
   (p11:sp1 ph8:r):f1
   4u pl1:f1
   p16:gp1*3
   d16
   4u BLKGRAD

   3m st cpmglist.inc
   lo to 2 times nbl

   3m cpmglist.res
   3m ipp4 ipp11 ipp31
   lo to 3 times ns

   6m mc #0 to 4
     F1QF()
     F2PH(calph(ph5, +90), caldel(d0, +in0))
exit


ph1=0
ph2=0
ph3=1
ph4=0 0 2 2
ph5=0

ph8=2
ph9=3
ph10=0
ph11=0 2

ph31=0 2 2 0
```

```
;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;pl62: f2 channel - a half of sp24 power level (in W)
;pl63: 0W
;sp1: f1 channel - shaped pulse 90 degree for water flip-back
;sp24: f2 channel - power level for 180 degree shaped pulse (adiabatic; 7kHz)
;spnam24: HS1_FC, HS4_FC, or HS8_FC
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p11: f1 channel -  90 degree shaped pulse for water flip-back  [1 msec]
;p16: homospoil/gradient pulse (1ms)
;p19: homospoil/gradient pulse (0.5ms)
;p20: homospoil/gradient pulse (0.2ms)
;p24: f2 channel -  180 degree adiabatic pulse   [4 msec]
;d0 : incremented delay (2D)                  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                             [30 msec]
;d16: delay for homospoil/gradient recovery
;cnst61: scaling factor (HS1: 0.295, HS4: 0.705, HS8: 0.836)
;l4: loop for R2rho delay. l4 may be 0,1,2,3,4,5
;l5: maximum value of l4
;inf2: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 4 * n
;ds: 16
;td1: number of experiments
;FnMODE: States-TPPI
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence



;for z-only gradients:
;gpz1: 10%
;gpz2: 6%
;gpz3: 50%
;gpz4: 10%
;gpz5: -20%
;gpz7: 25%


;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.50
;gpnam3: SMSQ10.100
;gpnam4: SINE.100
;gpnam5: SMSQ10.50
;gpnam7: SMSQ10.50



;$Id: SQ_methyl-HMQC_R2rho-3D.dc , 2020/04/02 F. Chao Exp $
```